*Replication Instructions for

*War-Related Victimization and Social Distance Toward Others: Evidence Following Russia's 2022 Invasion of Ukraine

*Sam Whitt, Douglas Page

*Below are instructions for replicating all manuscript and online appendix tables and figures in STATA. There are multiple replication datasets and do files, which are noted for each table and figure. Please contact Sam Whitt (swhitt@highpoint.edu) for questions regarding data replication. 

*Note: You may need to install STATA packages for the cibar, catcibar, and iebaltab commands. Use findit with the command name to identify and download the appropriate packets to install. 

*Note: In addition, some graphs require additional formatting using filename.grec files with the graph play command. To format a graph, simply run the command to generate the graph in the do file in STATA, then open the "Graph Editor" in STATA and click on the GREEN "Play Recording" button, then select "Browse" to select the grec file from the folder "grec files for STATA graph formatting" among Replication files. The name of the grec file is indicated in the note below the graph command in the do file for the specific graph you wish to format. This should automatically format the graph, which you may then save to a location of your choosing.

*Manuscript Replication

*"Stata user generated commands to install for replication purposes"

*"cibar"

ssc install cibar, replace

*"iebaltab from ietoolkit"

ssc install ietoolkit, replace

*"catcibar"

net install catcibar, from("https://aarondwolf.github.io/catcibar") replace


*In Text Replication up to robustness checks

*Use "CC Ukraine July 2022 replication data.dta"

*In terms of direct experiences with victimization, less than 1% report having been injured or knowing someone who was sexually assaulted, but 34% report having close friends or family injured or killed since February 24th, almost exclusively by Russian (91%) or unknown forces (7%).

tab injured 
tab sexassault  
tab faminjured 

*note: 91% comes from 596 out of 657 who indicate family or friends injured or killed by Russian forces. 7% comes from 45 out of 657 who indicated family or friends injured/killed by unknown forces.

*Nearly 12% report that their home or property was damaged or destroyed since the war started, while 14% report being internally displaced, with 4% displaced from newly occupied territories. 

tab homedestroyed

*note: 12% comes from 233 out of 2000 who report property destruction. 

tab moved
tab occupied 

*In our survey, 94% of the sample identifies as Ukrainian by `nationality', though only 73% generally prefer speaking in Ukrainian while 20% favor Russian.
 
tab nationality
tab language

*Finally, our occupation variable includes limited active-duty military personnel (2.3%) because they are currently deployed and not participating in telephone surveys. 

tab occupation

*Average feelings of closeness are much higher for ethnic Ukrainians than ethnic Russians (paired t-test = 55.3, p<0.0000) and even greater for citizens of Ukraine compared to citizens of Russia (paired t-test  = 133.7, p<0.0000). 

ttest ethnicuk = ethnicru
ttest citizenuk = citizenru

*On average, respondents rate ethnic Ukrainians five points higher on closeness than ethnic Russians within Ukraine, and 79% feel closer to ethnic Ukrainians than ethnic Russians. 

sum ethnicuk
sum ethnicru

*Code for generating ethnic bias dependent variable (already generated)
*gen magukbias = ethnicuk-ethnicru
*gen dukbias = 1 if magukbias>0
*replace dukbias = 0 if magukbias<1
*replace dukbias = . if magukbias==.
*tab dukbias

*In terms of social distance across citizenship, Ukrainian citizens are placed 8 points higher on closeness than citizens of Russia, and 96% feel closer to Ukrainian than Russian citizens. 

sum citizenuk
sum citizenru

*Code for generating citizen bias dependent variable (already generated)
*gen magcitizenbias = citizenuk-citizenru
*gen dcitizenbias = 1 if magcitizenbias>0
*replace dcitizenbias = 0 if magcitizenbias<1
*replace dcitizenbias =. if magcitizenbias==.
*tab dcitizenbias

*Figure 3.1 illustrates the distribution of responses on the 11-point scale where the modal response for closeness to ethnic Ukrainians is 10=very close (65%) and closeness to ethnic Russians is 0=not close at all (33.6%).

tab ethnicuk
tab ethnicru

*Figure 3.2 takes the difference in closeness to ethnic Ukrainians minus ethnic Russians. The distributions are skewed heavily toward a Ukrainian bias (scores greater than zero; 79%). In contrast, only a small minority report no difference in closeness across ethnicity (score equal to zero; 18%), while very few feel closer to ethnic Russians than Ukrainians (scores less than zero; 2.6%). 

tab dukbias
tab magukbias

*Differences are starker in Figure 3.3 when comparing social distance toward citizens of Ukraine (81% very close) relative to citizens of Russia (69% not close at all). 

tab citizenuk
tab citizenru

*Figure 3.4 shows the data are heavily skewed toward extreme polarization in closeness to Ukrainian citizens and distance toward citizens of Russia (96% report feeling closer to Ukrainian citizens than Russian citizens with only 3% indicating equalitarian feelings of closeness and <1% feel closer to Russian citizens than Ukrainian citizens). 

tab dcitizenbias
tab magcitizenbias

*Though few respondents reported personal injury or sexual assault (1%), one-third (34%) indicated that family or close friends had been killed, 14% had moved or resettled since the war began, and 12% indicated that their homes or property had been damaged or destroyed in fighting since Russia's February 2022 invasion. 

tab injured 
tab sexassault  
tab faminjured 
tab homedestroyed

*We focus on victimization experiences at the hands of Russian forces exclusively in our analysis because few respondents reported victimization by Ukrainian forces or unknown perpetrators (<4%). In contrast, 36.3% indicated some form of direct victimization by Russian forces. Our measure of direct victimization experiences consists of an additive index of responses to victimization at the hands of Russian forces, which is coded 0 for 63.7% who report no victimization experiences, coded 1 for 32.3% who report 1 experience, coded 2 for 3.7% who report 2 experiences, and 3 for 0.7% who report 3 distinct experiences.

tab addrvictimindex
tab drvictimindex

*…the effect size is relatively small for reducing both distance between Ukrainians and Russians by ethnicity (mean difference with no priming=5.17, SD=3.88, mean with priming=4.83, SD=3.86, two-sample t-test = -1.88, p<0.03, Cohen's d = 0.09)…

sum magukbias if victimorder==0
sum magukbias if victimorder==1
ttest magukbias, by(victimorder) unpaired unequal
esize twosample magukbias, by(victimorder)

*…and citizenship (mean difference with no priming=8.45, SD=2.61, mean difference with priming=8.21, SD=2.85, two-sample t-test = -1.94, p<0.03, Cohen's d = 0.09).

sum magcitizenbias if victimorder==0
sum magcitizenbias if victimorder==1
ttest magcitizenbias, by(victimorder) unpaired unequal
esize twosample magcitizenbias, by(victimorder)

*Also, in contrast to our preregistered expectations, we find no significant effects of victimization experience or interactions between our victimization prime and direct victimization experience on either ethnic bias (joint Wald F-test = 1.14, p<0.35) or citizenship-based bias (joint Wald F-test = 1.39, p<0.23).

reg magukbias victimorder##i.addrvictimindex , robust
contrast i.victimorder@i.addrvictimindex, effects
reg magcitizenbias victimorder##i.addrvictimindex , robust
contrast i.victimorder@i.addrvictimindex, effects

*Note: for continuous victimization index, see also
reg magukbias victimorder##c.addrvictimindex , robust
testparm i.victimorder##c.addrvictimindex
reg magcitizenbias victimorder##c.addrvictimindex , robust
testparm i.victimorder##c.addrvictimindex

*Figure 1

*Figure 1 (use Data_OMN_Nov_2021_KIIS_ukr.dta and do file)

*Figure 2

*Figure 2 (use Data_OMN_Kiis_prepostwar.dta and do file)

*Table 2

sum ethnicuk ethnicru magukbias dukbias
sum citizenuk citizenru magcitizenbias dcitizenbias

*Figure 3

graph twoway (histogram ethnicuk, discrete percent) (histogram ethnicru, discrete percent)
graph save g1
histogram magukbias, discrete percent
graph save g2
graph combine "g1.gph" "g2.gph"
*Note additional formatting requires the "Figure 3.1-2 formatting.grec" file with the command graph play "Figure 3.1-2 formatting.grec" 

graph twoway (histogram citizenuk, discrete percent) (histogram citizenru, discrete percent)
graph save g3
histogram magcitizenbias, discrete percent
graph save g4
graph combine "g3.gph" "g4.gph"
*Note additional formatting requires the "Figure 3.3-4 formatting.grec" file with the command graph play "Figure 3.3-4 formatting.grec" 

*Table 3

reg magukbias victimorder##c.addrvictimindex, robust
vif
reg magukbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
vif
reg magcitizenbias victimorder##c.addrvictimindex, robust
vif
reg magcitizenbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
vif

*Table 4

reg revethnicuk victimorder, robust
reg revethnicru victimorder, robust
reg revcitizenuk victimorder, robust
reg revcitizenru victimorder, robust

*Robustness Check – Contextualizing Identity

*Use CC Ukraine Jan 2023 replication data.dta and do file

*Figure 4

*Use "CC Ukraine Jan 2023 Figure 4 data.dta" and enter code
*cibar closeall, over1(identitytxt)

*Sample Demographics (July 2022)

sum victimorder i.injured i.faminjured i.sexassault i.homedestroyed moved occupied female age education i.nationality i.language i.surveylang i.occupation income rural i.region4

*Robustness Checks

*Manuscript Table 3 (Tobit Regression)

tobit magukbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, ll ul

tobit magcitizenbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, ll ul

*Robustness Check: Binary Victimization Measurement

reg magukbias victimorder##drvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
reg magcitizenbias victimorder##drvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust

*Robustness Check – Ordinal Measure of Victimization

tab addrvictimindex
reg magukbias victimorder##i.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
reg magcitizenbias victimorder##i.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust

*Moderated Effects of Victimization Experience on Social Distance (OLS Regression)

reg magukbias victimorder##i.addrvictimindex , robust
margins victimorder, at(addrvictimindex=(0 (1) 3))
marginsplot, yline(0)
graph save g5
reg magcitizenbias victimorder##i.addrvictimindex , robust
margins victimorder, at(addrvictimindex=(0 (1) 3))
marginsplot, yline(0)
graph save g6
graph combine "g5.gph" "g6.gph"

reg magukbias victimorder##i.addrvictimindex , robust
margins, dydx(victimorder) at(addrvictimindex=(0 (1) 3))
marginsplot, yline(0)
graph save g7
reg magcitizenbias victimorder##i.addrvictimindex , robust
margins, dydx(victimorder) at(addrvictimindex=(0 (1) 3))
marginsplot, yline(0)
graph save g8
graph combine "g7.gph" "g8.gph"

*Moderated Effects of Victimization Experience on Social Distance (Interflex Estimates)

interflex magukbias victimorder addrvictimindex, vce(robust)
graph save g9
interflex magcitizenbias victimorder addrvictimindex, vce(robust)
graph save g10
graph combine "g9.gph" "g10.gph"

*Balance Tests on Victimization Treatment 
ksmirnov rinjured, by(victimorder)
ksmirnov rfaminjured, by(victimorder)
ksmirnov rsexassault, by(victimorder)
ksmirnov rhomedestroyed, by(victimorder)
ksmirnov moved, by(victimorder)
ksmirnov occupied, by(victimorder)
ksmirnov russian, by(victimorder)
ksmirnov russpeaker, by(victimorder)
ksmirnov surveylang, by(victimorder)
ksmirnov female, by(victimorder)
ksmirnov age, by(victimorder)
ksmirnov education, by(victimorder)
ksmirnov income, by(victimorder)
ksmirnov region4, by(victimorder)
ksmirnov rural, by(victimorder)

*note: see also 
iebaltab rinjured rfaminjured rsexassault moved occupied female age education russian russpeaker occupation income rural region4, groupvar(victimorder) savexlsx(victimbalance)

*Victimization Treatment Effects Sensitivity Analysis

reg magukbias victimorder##c.addrvictimindex, robust
reg magukbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
reg magcitizenbias victimorder##c.addrvictimindex, robust
reg magcitizenbias victimorder##c.addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust
regsensitivity bounds magukbias victimorder addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust plot
regsensitivity bounds magcitizenbias victimorder addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, robust plot
regsensitivity bounds magukbias victimorder addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, oster robust plot
regsensitivity bounds magcitizenbias victimorder addrvictimindex ethnicorder i.rinjured i.rfaminjured i.rsexassault moved occupied female age education russian russpeaker i.occupation income urban_rural i.region4, oster robust plot

*Intercorrelation among Victimization Items 

pwcorr rinjured rfaminjured rsexassault rhomedestroyed moved occupied, sig

*Factor Analysis of Victimization-Related Items

factor rinjured rfaminjured rsexassault rhomedestroyed moved occupied

*Correlates of Victimization (Logit Regression)

logit rinjured  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)
logit rfaminjured  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)
logit rsexassault  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)
logit rhomedestroyed  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)
logit moved  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)
logit occupied  female age education russian russpeaker i.occupation income urban_rural i.region4 date, cluster(oblast)

*Changes in Social Distance Over Time (July 2022-January 2023)

*Use "CC Ukraine 2022-2023 combined data.dta and see READ ME file for code

*Sample Demographics (Follow-up January 3-11, 2023 survey)

*Use "CC Ukraine Jan 2023 replication data.dta"

*Balance Tests (Follow-up January 3-11, 2023 survey Identity Treatments)

*Use "CC Ukraine Jan 2023 replication data.dta"


*Power Calculations 

power oneway, ngroups(2) n1(1011) n2(989) power(0.80 0.90 0.95 0.99)

*Effect sizes (Victimization Priming)
esize twosample magukbias, by(victimorder)
esize twosample magcitizenbias, by(victimorder)

*Effect sizes (Identity Priming)

*Use "CC Ukraine Jan 2023 replication data.dta"

log close